;;;
;;;	Author: Henry C. Francis
;;;		425 N. Ashe St.
;;;		Southern Pines, NC 28387
;;;
;;;	http://www.paracadd.com
;;;	All rights reserved.
;;;
;;;	Copyright:
;;;	   Edited:
;;;
(defun C:DIMEXT ()
  (vl-load-com)
  (if leader-option
    NIL
    (setq leader-option "Both")
  ) ;_ end of IF
  (setq	leader-option
	 (ukword
	   1
	   "1st 2nd 3rd One Two Both All First Second None"
	   "Specify extension line(s) to turn off (None turns them both on)"
	   leader-option
	 ) ;_ end of ukword
  ) ;_ end of SETQ
  (WHILE
  (setq selected-dim (entsel))
  (if (and selected-dim
	   (eq (cdr (assoc 0 (entget (car selected-dim)))) "DIMENSION")
      ) ;_ end of AND
    (progn
      (setq dimobj (vlax-ename->vla-object (car selected-dim)))
;;;(vlax-dump-object dimobj) ; for inspecting a selection
      (cond
	((or
	   (eq leader-option "1st")
	   (eq leader-option "One")
	   (eq leader-option "First")
	 ) ;_ end of OR
	 (vlax-put-property dimobj 'ExtLine1Suppress -1) ; -1 = off
	 (vlax-put-property dimobj 'ExtLine2Suppress 0) ; 0 = on
	)
	((or
	   (eq leader-option "2nd")
	   (eq leader-option "Two")
	   (eq leader-option "Second")
	 ) ;_ end of OR
	 (vlax-put-property dimobj 'ExtLine1Suppress 0) ; -1 = off
	 (vlax-put-property dimobj 'ExtLine2Suppress -1) ; 0 = on
	)
	((or
	   (eq leader-option "3rd")
	   (eq leader-option "Both")
	   (eq leader-option "All")
	 ) ;_ end of OR
	 (vlax-put-property dimobj 'ExtLine1Suppress -1) ; -1 = off
	 (vlax-put-property dimobj 'ExtLine2Suppress -1) ; 0 = on
	)
	((eq leader-option "None")
	 (vlax-put-property dimobj 'ExtLine1Suppress 0) ; -1 = off
	 (vlax-put-property dimobj 'ExtLine2Suppress 0) ; 0 = on
	)
      ) ;_ end of COND
    ) ;_ end of PROGN
    (progn
      (princ "\nNO DIMENSION WAS SELECTED! ")
      (princ)
    ) ;_ end of PROGN
  ) ;_ end of IF
  )
  (princ)
) ;_ end of DEFUN
(DEFUN C:DIMEXTOFF () (C:DIMEXT))
(PRINC)